Differential Geometry in Cartesian Coordinates

Suppose we have a vector

```
In[3]:= r = {x, y, f[x, y]}
Out[3]= {x, y, f[x, y]}
```

The Cartesian derivatives of this vector are

```
In[4]:= rx = D[r, x]
        ry = D[r, y]
```

Out[4]= $\{1, 0, f^{(1,0)}[x, y]\}$

Out[5]= $\{0, 1, f^{(0,1)}[x, y]\}$

Examine the dot products of these derivatives

```
In[6]:= E0 = rx.rx
        F0 = rx.ry
        G0 = ry.ry
```

Out[6]= $1 + f^{(1,0)}[x, y]^2$

Out[7]= $f^{(0,1)}[x, y]\, f^{(1,0)}[x, y]$

Out[8]= $1 + f^{(0,1)}[x, y]^2$

These dot products are the coefficients E, F, G of the First Fundamental Form. Recall that $EG - F^2$ appears throughout are differential geometry calculations

```
In[9]:= rad = Simplify[E0 * G0 - F0^2]
```

Out[9]= $1 + f^{(0,1)}[x, y]^2 + f^{(1,0)}[x, y]^2$

The vectors rx and ry lie in the tangent plane associated with the point on the surface. Consequently, the normal vector n to the surface at r is just the cross product of rx and ry since they both lie in the tangent plane.

```
In[10]:= n = Cross[rx, ry]
```

Out[10]= $\{-f^{(1,0)}[x, y], -f^{(0,1)}[x, y], 1\}$

The magnitude of n is

```
In[11]:= Sqrt[n.n]
```

Out[11]= $\sqrt{1 + f^{(0,1)}[x, y]^2 + f^{(1,0)}[x, y]^2}$

so the unit normal is

In[67]:= **nhat = n / Sqrt[n.n]**

Out[67]= $$\left\{-\frac{f^{(1,0)}[x,y]}{\sqrt{1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2}},\ -\frac{f^{(0,1)}[x,y]}{\sqrt{1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2}},\ \frac{1}{\sqrt{1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2}}\right\}$$

The local curvature of the surface at the point r is described by how quickly and it what direction the unit normal rotates for small changes in position. This rate of change is given by the derivatives of the unit normal.

In[13]:= **nhatx = D[nhat, x]**
nhaty = D[nhat, y]

Out[13]= $$\left\{-\frac{f^{(2,0)}[x,y]}{\sqrt{1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2}}+\frac{f^{(1,0)}[x,y]\left(2f^{(0,1)}[x,y]f^{(1,1)}[x,y]+2f^{(1,0)}[x,y]f^{(2,0)}[x,y]\right)}{2\left(1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2\right)^{3/2}},\right.$$
$$-\frac{f^{(1,1)}[x,y]}{\sqrt{1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2}}+\frac{f^{(0,1)}[x,y]\left(2f^{(0,1)}[x,y]f^{(1,1)}[x,y]+2f^{(1,0)}[x,y]f^{(2,0)}[x,y]\right)}{2\left(1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2\right)^{3/2}},$$
$$\left.-\frac{2f^{(0,1)}[x,y]f^{(1,1)}[x,y]+2f^{(1,0)}[x,y]f^{(2,0)}[x,y]}{2\left(1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2\right)^{3/2}}\right\}$$

Out[14]= $$\left\{-\frac{f^{(1,1)}[x,y]}{\sqrt{1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2}}+\frac{f^{(1,0)}[x,y]\left(2f^{(0,1)}[x,y]f^{(0,2)}[x,y]+2f^{(1,0)}[x,y]f^{(1,1)}[x,y]\right)}{2\left(1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2\right)^{3/2}},\right.$$
$$-\frac{f^{(0,2)}[x,y]}{\sqrt{1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2}}+\frac{f^{(0,1)}[x,y]\left(2f^{(0,1)}[x,y]f^{(0,2)}[x,y]+2f^{(1,0)}[x,y]f^{(1,1)}[x,y]\right)}{2\left(1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2\right)^{3/2}},$$
$$\left.-\frac{2f^{(0,1)}[x,y]f^{(0,2)}[x,y]+2f^{(1,0)}[x,y]f^{(1,1)}[x,y]}{2\left(1+f^{(0,1)}[x,y]^2+f^{(1,0)}[x,y]^2\right)^{3/2}}\right\}$$

If these vectors are projected onto the derivatives of the position vector, then the coefficients of the Second Fundamental Form are obtained (with a minus sign as the sign convention).

In[58]:=
```
L0 = -nhatx.rx
M0 = -(1/2)*(nhatx.ry + nhaty.rx)
N0 = -nhaty.ry
```

Out[58]= $\dfrac{f^{(2,0)}[x, y]}{\sqrt{1 + f^{(0,1)}[x, y]^2 + f^{(1,0)}[x, y]^2}}$

Out[59]= $\dfrac{f^{(1,1)}[x, y]}{\sqrt{1 + f^{(0,1)}[x, y]^2 + f^{(1,0)}[x, y]^2}}$

Out[60]= $\dfrac{f^{(0,2)}[x, y]}{\sqrt{1 + f^{(0,1)}[x, y]^2 + f^{(1,0)}[x, y]^2}}$

Since rx and ry are in the tangent plane and nhat is perpendicular to the tangent plane, the dot products rx . nhat = ry . nhat = 0. Differentiating these expresions gives

$(r_x . \text{nhat})_x = r_x . \text{nhat}_x + r_{xx} . \text{nhat} = 0 \rightarrow r_{xx} . \text{nhat} = - r_x . \text{nhat}_x$

$(r_x . \text{nhat})_y = r_x . \text{nhat}_y + r_{xy} . \text{nhat} = 0 \rightarrow r_{xy} . \text{nhat} = - r_x . \text{nhat}_y$

$(r_y . \text{nhat})_x = \text{ry} . \text{nhat}_x + r_{xy} . \text{nhat} = 0 \rightarrow r_{xy} . \text{nhat} = -r_y . \text{nhat}_x$

$(r_y . \text{nhat})_y = \text{ry} . \text{nhat}_y + r_{yy} . \text{nhat} = 0 \rightarrow r_{yy} . \text{nhat} = -r_y . \text{nhat}_y$

Consequently, the coefficients of the Second Fundamental Form can also be written as

In[68]:=
```
rxx = D[rx, x]
rxy = D[rx, y]
ryy = D[ry, y]
L0 = rxx.nhat
M0 = rxy.nhat
N0 = ryy.nhat
```

Out[68]= $\{0, 0, f^{(2,0)}[x, y]\}$

Out[69]= $\{0, 0, f^{(1,1)}[x, y]\}$

Out[70]= $\{0, 0, f^{(0,2)}[x, y]\}$

Out[71]= $\dfrac{f^{(2,0)}[x, y]}{\sqrt{1 + f^{(0,1)}[x, y]^2 + f^{(1,0)}[x, y]^2}}$

Out[72]= $\dfrac{f^{(1,1)}[x, y]}{\sqrt{1 + f^{(0,1)}[x, y]^2 + f^{(1,0)}[x, y]^2}}$

Out[73]= $\dfrac{f^{(0,2)}[x, y]}{\sqrt{1 + f^{(0,1)}[x, y]^2 + f^{(1,0)}[x, y]^2}}$

Differential Geometry with Parametrized Surfaces

We can also generalize this technique and define the vector r as follows

$r = [g(u, v), h(u, v), f[g(u, v), h(u, v)]]$

where u and v are called parameters and the functions g() and h() describe how the x and y coordinates of r vary as u and v are changed. The coefficients of the Fundamental Forms are now defined as

$E = r_u . r_u$
$F = r_u . r_v$
$G = r_v . r_v$

$L = -r_u . \text{nhat}_u = r_{\text{uu}} . \text{nhat}$
$M = -\left(\frac{1}{2}\right)(r_u . \text{nhat}_u + r_v . \text{nhat}_v) = r_{\text{uv}} . \text{nhat}$
$N = -r_v . \text{nhat}_v = r_{\text{uv}} . \text{nhat}$

With these definitions, the Gaussian, Mean and Principal Curvatures are all still calculated in the same manner. The advantage of this technique is the coordinate systems or parameterizations that are conducive to the calculation at hand can be used. A few examples of these parameterizations are

Cartesian Coordinates
u = x ; v = y and g(x,y) = x; h(x,y) = y; f(g(x,y), h(x,y)) = f(x, y)

Polar Coordinates
u = r ; v = θ and g(r, θ) = r cosθ; h(r,θ) = r sinθ; f(g(r,θ), h(r,θ)) = f(r,θ)

Differential Geometry with Rotationally Symmetric Functions

Here we will use x1 for the vector to the point on the surface to avoid confusion with the radial coordinate r. Since we are assuming rotational symmetry, f(r,θ) = f(r). This is a general point on the rotationally symmetric function

```
In[75]:= Clear[r]

In[76]:= x1 = {r * Cos[θ], r * Sin[θ], f[r]}

Out[76]= {r Cos[θ], r Sin[θ], f[r]}
```

Calculate the first and second derivatives

```
In[77]:= x1r = D[x1, r]

Out[77]= {Cos[θ], Sin[θ], f′[r]}

In[78]:= x1θ = D[x1, θ]

Out[78]= {-r Sin[θ], r Cos[θ], 0}

In[79]:= x1rr = D[x1r, r]

Out[79]= {0, 0, f′′[r]}
```

In[80]:= `x1rθ = D[x1r, θ]`

Out[80]= `{-Sin[θ], Cos[θ], 0}`

In[81]:= `x1θθ = D[x1θ, θ]`

Out[81]= `{-r Cos[θ], -r Sin[θ], 0}`

Calculate the First Fundmental Form

In[82]:= `E0 = Simplify[x1r.x1r]`

Out[82]= $1 + f'[r]^2$

In[83]:= `F0 = Simplify[x1r.x1θ]`

Out[83]= 0

In[84]:= `G0 = Simplify[x1θ.x1θ]`

Out[84]= r^2

Calculate the unit normal

In[87]:= `n = Simplify[Cross[x1r, x1θ]]`

Out[87]= `{-r Cos[θ] f'[r], -r Sin[θ] f'[r], r}`

In[89]:= `nhat = Simplify[n / Sqrt[n.n]]`

Out[89]= $$\left\{-\frac{r\,\text{Cos}[\theta]\,f'[r]}{\sqrt{r^2\left(1+f'[r]^2\right)}},\ -\frac{r\,\text{Sin}[\theta]\,f'[r]}{\sqrt{r^2\left(1+f'[r]^2\right)}},\ \frac{r}{\sqrt{r^2\left(1+f'[r]^2\right)}}\right\}$$

In[90]:= $$\text{nhat} = \left\{-\frac{\text{Cos}[\theta]\,f'[r]}{\sqrt{1+(f'[r])^2}},\ -\frac{\text{Sin}[\theta]\,f'[r]}{\sqrt{1+(f'[r])^2}},\ \frac{1}{\sqrt{1+(f'[r])^2}}\right\}$$

Out[90]= $$\left\{-\frac{\text{Cos}[\theta]\,f'[r]}{\sqrt{1+f'[r]^2}},\ -\frac{\text{Sin}[\theta]\,f'[r]}{\sqrt{1+f'[r]^2}},\ \frac{1}{\sqrt{1+f'[r]^2}}\right\}$$

Calculate the Second Fundmental Form

In[91]:= `L0 = Simplify[x1rr.nhat]`

Out[91]= $$\frac{f''[r]}{\sqrt{1+f'[r]^2}}$$

In[92]:= `M0 = Simplify[x1rθ.nhat]`

Out[92]= 0

In[93]:= `N0 = Simplify[x1θθ.nhat]`

Out[93]= $$\frac{r\,f'[r]}{\sqrt{1+f'[r]^2}}$$

The Gaussian Curvature K0 is given by

In[94]:= `K0 = (L0 * N0 - M0^2) / (E0 * G0 - F0^2)`

Out[94]= $$\frac{f'[r]\, f''[r]}{r\left(1+f'[r]^2\right)^2}$$

The Mean Curvature H0 is given by

In[95]:= `H0 = Simplify[(E0 * N0 - 2 * F0 * M0 + G0 * L0) / (2 * (E0 * G0 - F0^2))]`

Out[95]= $$\frac{f'[r] + f'[r]^3 + r\, f''[r]}{2\, r\left(1+f'[r]^2\right)^{3/2}}$$

In[96]:= `κ1 = Simplify[H0 + Sqrt[H0^2 - K0]]`
`κ2 = Simplify[H0 - Sqrt[H0^2 - K0]]`

Out[96]= $$\frac{1}{2}\left(\sqrt{\frac{\left(f'[r] + f'[r]^3 - r\, f''[r]\right)^2}{r^2\left(1+f'[r]^2\right)^3}} + \frac{f'[r] + f'[r]^3 + r\, f''[r]}{r\left(1+f'[r]^2\right)^{3/2}}\right)$$

Out[97]= $$\frac{1}{2}\left(-\sqrt{\frac{\left(f'[r] + f'[r]^3 - r\, f''[r]\right)^2}{r^2\left(1+f'[r]^2\right)^3}} + \frac{f'[r] + f'[r]^3 + r\, f''[r]}{r\left(1+f'[r]^2\right)^{3/2}}\right)$$

In[98]:= $$\kappa 1 = \frac{1}{2}\left(\text{Together}\left[\frac{f'[r] + f'[r]^3 - r\, f''[r]}{r\left(1+f'[r]^2\right)^{3/2}} + \frac{f'[r] + f'[r]^3 + r\, f''[r]}{r\left(1+f'[r]^2\right)^{3/2}}\right]\right)$$

Out[98]= $$\frac{f'[r]}{r\sqrt{1+f'[r]^2}}$$

In[99]:= $$\kappa 2 = \frac{1}{2}\left(\text{Together}\left[-\frac{f'[r] + f'[r]^3 - r\, f''[r]}{r * \left(1+f'[r]^2\right)^{3/2}} + \frac{f'[r] + f'[r]^3 + r\, f''[r]}{r\left(1+f'[r]^2\right)^{3/2}}\right]\right)$$

Out[99]= $$\frac{f''[r]}{\left(1+f'[r]^2\right)^{3/2}}$$

In[100]:= `θ0 = 2 * M0 * Sqrt[L0 * N0] / (L0 * N0 - E0 * G0)`

Out[100]= 0